
Principe :

Le cadran d'appel téléphonique est un dispositif qui envoie des impulsions 
d'ouverture de contact en série.
Le chiffre 1 correspond à une impulsion jusqu'au chiffre 0 pour 10 impulsions. 
La fréquence est de 10 impulsions par seconde, ce qui équivaut à une fréquence 
de 10 Hz.
La durée d'une impulsion est donc de 100 millisecondes, avec environ 66 ms pour 
ouvrir le contact et 33 ms pour le fermer.
Le rapport ouverture/fermeture est donc de 2 = 66/33

Ici, le code dans l'Arduino Uno compte 
les impulsions d'ouverture du contact et 
il les affichera sur le Moniteur série 
de Arduino IDE en 9600 bit/s

Le Code

Le code disponible en téléchargement sur :  http://jacques.reumont.free.fr/Arduino/Test_Cadran_Telephone.zip
  

http://jacques.reumont.free.fr/
http://jacques.reumont.free.fr/Arduino/Test_Cadran_Telephone.zip


/* Test_Cadran_Telephone jacques.reumont@orange.fr        http://jacques.reumont.free.fr
  ___             _                       _____         _                   _
/  __| __ _  __| | _ _  __ _ _  _   |_  _|  ___  | |   ___  _ __  | |_     ___   _ _   ___
| (__ / _` |/ _` || '_ |/ _` ||  '  \   |  |  /  - _) | | / -_ ) | '_ \ | '   \  / _  \ | '  \ / -_ )
\___|\__,_|\__,_||_|  \__,_||_||_|   |_|  \___| |_| \___ || .__/ |_||_| \___ /|_||_|\___|
                                                                        |_|
*/
#define VERSION "Test_Cadran_Telephone J.R" // Version Février 2025
#define pinPulse 2                  // pin Entrée fil Rouge pulse du Cadran (contact NF du cadran)
                                                // le fil Rouge/blanc du Cadran au GND de l'arduino
#define pinLed 13                   // pin pinLed sur la carte arduino
//*** Déclaration des variables globales ***
byte chiffre_Cadran = 0 ;         // nombre de pinPulse lors du retour de la rotation du cadran
boolean m_Pulse = 0 ;             // Mémoire pinPulse niveau haut/bas du cadran
unsigned long time_now = 0 ; // pour la tempo
int tempoOK = 300 ;                // tempo fin des pulses du cadran
int tempoKO = 1000 ;              // tempo cadran absent
boolean m_OK = 0 ;                 // Mémoire ok du numéro

//       _________
//___/   SETUP   \____________________________________________________
//
void setup(){

    Serial.begin(9600);                         // Vitesse du terminal série USB pour info sur la console
    Serial.println(VERSION);                  // Info version sur le terminal série USB vitesse 9600
    pinMode(pinPulse, INPUT_PULLUP); // pin en entrée: Pulse du cadran du Téléphone
    pinMode(pinLed, OUTPUT) ;            // pin Sortie Led de l'Arduino , info présence du cadran
}
//       ________
//___/   LOOP   \____________________________________________________
//
void loop() {
    // Comptage des Pulses du cadran: contact ouvert = 66ms ; contact fermé = 33ms
    // si le contact du cadran est fermé (au repos) maxi 33ms
    if (!digitalRead(pinPulse) && !m_Pulse ) {delay(10); }     // anti-rebond des Pulses du cadran
    if (!digitalRead(pinPulse) && !m_Pulse ) {
        // visu du défilement du n° pour le reglage du contact du cadran
        if (chiffre_Cadran){Serial.print(chiffre_Cadran);}
        m_Pulse=1;
        digitalWrite(pinLed, 1);
        delay(15);
    }
    
    // si le contact du cadran est ouvert ou non connecté maxi 66ms
    if (digitalRead(pinPulse) && m_Pulse ) {delay(10); }       // anti-rebond du cadran
    if (digitalRead(pinPulse) && m_Pulse ) {
        time_now = millis() ; // raz de la tempo
        m_Pulse = 0 ;
        m_OK = 1 ;
        chiffre_Cadran++ ;
        digitalWrite(pinLed, 0);
        delay(30);
    }

    // résultat KO
    if(digitalRead(pinPulse) && m_OK && (millis() - time_now) > tempoKO){
        Serial.println(" ?");
        chiffre_Cadran = 0 ;
        m_OK = 0 ;
    }

    // résultat OK
    if(!digitalRead(pinPulse) && m_OK && (millis() - time_now) > tempoOK){
        m_OK = 0;
        Serial.print(" -> OK Chiffre: ");
        Serial.println(chiffre_Cadran%10);
        chiffre_Cadran = 0;
    }

}
//      ____________
//___/     FIN          \___________________________________________________
//

